

FIELLER' S THEOREM
Theory and Application
With Computations in SAS

© Copyright 2001 by Mary A. Marion

**FIELLER'S THEOREM -Theory and Application
With Computations in SAS**

© Copyright 2001 by Mary A. Marion

Introduction

Confidence intervals for ratio estimators can be computed using Fieller's Theorem. The enclosed two macros written in SAS compute a $(1-\alpha) \times 100\%$ confidence interval for such estimators.

Theory

The theory behind Fieller's fiducial intervals as presented by Dr. George Milliken of Kansas State University and Dr. John Miller of George Mason University is as follows. Consider the ratio $p = \alpha / \beta$ and its estimator

$m=a/b$ where $\begin{pmatrix} a \\ b \end{pmatrix}$ follows a bivariate normal distribution with mean $\begin{pmatrix} \alpha \\ \beta \end{pmatrix}$ and covariance $\sigma_x^2 \begin{pmatrix} v_{11} & v_{12} \\ v_{12} & v_{22} \end{pmatrix}$.

$$\text{Let } d = a - bp = \begin{bmatrix} 1 & -p \end{bmatrix} \begin{bmatrix} a \\ b \end{bmatrix}. \quad E(d) = \alpha - p\beta = \alpha - \frac{\alpha}{\beta} \beta = 0$$

$$\text{Var}(d) = \text{Var}(a) + p^2 \text{Var}(b) - 2p \text{Cov}(a, b) = \sigma^2(v_{11} - 2pv_{12} + p^2v_{22})$$

$$d \sim N \left\{ d, \begin{bmatrix} 1 & -p \end{bmatrix} \begin{bmatrix} \alpha \\ \beta \end{bmatrix}, \sigma^2 \begin{bmatrix} 1 & -p \end{bmatrix} \begin{bmatrix} v_{11} & v_{12} \\ v_{12} & v_{22} \end{bmatrix} \begin{bmatrix} 1 \\ -p \end{bmatrix} \right\}$$

$$\sim N \left\{ d, \alpha - p\beta, \sigma^2(v_{11} - 2pv_{12} + p^2v_{22}) \right\}$$

Now form a t statistic since σ^2 is unknown.

$$t_c = \frac{d}{\hat{\sigma} \sqrt{v_{11} - 2pv_{12} + p^2v_{22}}} \sim t_{v, \alpha/2}$$

The $(1-\alpha) \times 100\%$ confidence interval about p is that set of values of p such that

$$P \left[-t_{\alpha/2, v} \leq t_c \leq t_{\alpha/2, v} \right] = 1 - \alpha \quad \text{or} \quad P \left[t_c^2 \leq t_{\alpha/2, v}^2 \right] = 1 - \alpha$$

I need to find all p such that

$$t_c^i \leq t_{\alpha/i,v}^i \rightarrow \frac{(a - bp)^i}{\hat{\sigma}^i(v_{11} - 2pv_{12} + p^2v_{22})} \leq t_{\alpha/i,v}^i \rightarrow$$

$$(a - bp)^i \leq t_{\alpha/i,v}^i \hat{\sigma}^i(v_{11} - 2pv_{12} + p^2v_{22}) \rightarrow$$

$$a^i - 2abp + b^2p^2 \leq t_{\alpha/i,v}^i s^i(v_{11} - 2pv_{12} + p^2v_{22}) \rightarrow$$

$$\frac{a^i}{b^i} - 2\frac{a}{b}p + p^2 \leq \left(\frac{ts}{b}\right)^i (v_{11} - 2pv_{12} + p^2v_{22}) \rightarrow$$

$$\left(\frac{a}{b} - p\right)^i \leq k^i(v_{11} - 2pv_{12} + p^2v_{22}) \rightarrow (m - p)^i \leq k^i(v_{11} - 2pv_{12} + p^2v_{22}) \rightarrow$$

$$m^i - 2mp + p^2 \leq k^iv_{11} - 2k^ipv_{12} + k^ip^2v_{22} \rightarrow$$

$$p^2(1 - k^2v_{22}) + p(2k^2v_{12} - 2m) + (m^2 - k^2v_{11}) \leq 0 \rightarrow$$

$$p^2 A + p B + C \leq 0$$

Next solve $p^2A + pB + C = 0$ for p

$$\rightarrow p = \frac{-B \pm \sqrt{B^2 - 4AC}}{2A} \rightarrow$$

$$p = \frac{2m - 2k^2v_{12} \pm \sqrt{2^i(m - k^2v_{12})^2 - 4(1 - k^2v_{22})(m^2 - k^2v_{11})}}{2(1 - k^2v_{22})} \rightarrow$$

$$p = \frac{2(m - k^2v_{12}) \pm 2\sqrt{(m - k^2v_{12})^2 - (1 - k^2v_{22})(m^2 - k^2v_{11})}}{2(1 - k^2v_{22})}$$

Fieller's theorem states that upper and lower fiducial limits to p are

$$p = \frac{m - k^2v_{12} \pm \sqrt{m^2 - 2k^2v_{12}m + k^4v_{12}^2 - m^2 + k^2v_{22}m^2 + k^2v_{11} - k^4v_{11}v_{12}}}{(1 - k^2v_{22})}$$

The computations found in the popular text "Statistical Method in Biological Assay" by Finney are similar. In this text application of

Fieller's theorem computes the upper and lower fiducial limits to p as

$$m_L, m_U = \left[m - \frac{g v_{12}}{v_{22}} \pm \frac{ts}{b} \left\{ v_{11} - 2m v_{12} + m^2 v_{22} - g \left(v_{11} - \frac{v_{12}^2}{v_{22}} \right)^{1/2} \right\} / (1 - g) \right]$$

where $g = \frac{t^2 s^2 v_{22}}{b^2}$ and t is the t -deviate with v degrees of freedom.

Note that in the above calculations that

$$A = 1 - k^2 v_{22}, \quad B = 2 \left(k^2 v_{12} - m \right), \quad C = m^2 - k^2 v_{11}$$

$$k^2 = \left[\frac{ts}{b} \right]^2, \quad p = \alpha / \beta \quad \text{and} \quad m = a / b$$


By rewriting the equation for k , I can find an expression for t . Then compute the area corresponding to the minimum p -value for which an appropriate solution is possible.

$$k = \frac{\hat{\sigma} t_{\alpha/2, v}}{b} \rightarrow t_{\alpha/2, v} = \frac{kb}{\hat{\sigma}} = \frac{kb}{s}$$


Now compute $2(1 - T(kb/s))$ where T is the cdf of a t random variable. This is the minimum observed significance level which will yield an appropriate interval.

Four possible situations can occur in the solution of these equations.


Case 1a. $A > 0$ and $B^2 - 4AC > 0$ Two distinct solutions exist.




Case 1b. $A > 0$ and $B^2 - 4AC = 0$ Single solution (a single point) exists.




Case 2 $A > 0$ and $B^2 - 4AC < 0$ A confidence interval does not exist.



Case 3 $A < 0$ and $B^2 - 4AC > 0$. Inequality $p^2 A + p B + C \leq 0$ does not lead to a useful confidence region.



Case 4 $A < 0$ and $B^2 - 4AC < 0$. Confidence interval does not exist (may as well not collected any data at all).



All 4 situations are reported in the SAS macros. Details are presented under title "Interval Diagnostics".

Application.

This approach requires a simple linear regression equation of the form $y = a x + b$ where $\begin{bmatrix} \alpha \\ \beta \end{bmatrix}$ follows a bivariate normal distribution. Suppose further that a and b are unbiased estimates of α and β forming a linear function of a set of observations with normally distributed errors. An ANOVA of the data will give a error mean square, s^2 , with v degrees of freedom. The estimated variances of a , b , and their covariance may be expressed as $s^2 v_{11}$, $s^2 v_{22}$ and $s^2 v_{12}$.

The technique is used in calibration where the unknown is x . y is given by the experimenter and using the formula $y = \hat{\alpha} + \hat{\beta}x$, x is computed for various values of y .

First the joint distribution of y , $\hat{\alpha}$ and $\hat{\beta}$ is found.

$$\begin{pmatrix} Y_o \\ \hat{\alpha} \\ \hat{\beta} \end{pmatrix} \sim N \left(\begin{pmatrix} Y_o \\ \hat{\alpha} \\ \hat{\beta} \end{pmatrix}, \begin{pmatrix} \alpha + \beta x_o \\ \alpha \\ \beta \end{pmatrix}, \sigma^2 \begin{bmatrix} 1 & 0 \\ 0 & (X'X)^{-1} \end{bmatrix} \right)$$

Computation of a confidence interval for the ratio $x_o = \frac{y_o - \hat{\alpha}}{\hat{\beta}}$ can be simplified using matrix operations.

$$\begin{pmatrix} y_o - \hat{\alpha} \\ \hat{\beta} \end{pmatrix} = \begin{pmatrix} 1 & -1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} y_o \\ \hat{\alpha} \\ \hat{\beta} \end{pmatrix} \sim N \left(\begin{pmatrix} y_o - \hat{\alpha} \\ \hat{\beta} \end{pmatrix}, \begin{pmatrix} Bx_o \\ B \end{pmatrix}, \sigma^2 \underbrace{\begin{pmatrix} 1 & -1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{bmatrix} 1 & 0 \\ 0 & (X'X)^{-1} \end{bmatrix}}_{\begin{pmatrix} v_{11} & v_{12} \\ v_{12} & v_{22} \end{pmatrix}} \begin{pmatrix} 1 & 0 \\ -1 & 0 \\ 0 & 1 \end{pmatrix} \right)$$

Now apply Fieller's theorem with

$$a = y - \hat{\alpha}, \quad b = \hat{\beta}, \quad \hat{\sigma}^2 = \frac{\sum_1^n (y_i - \hat{\alpha} - \hat{\beta}x)^2}{n - 2}, \quad df = v = n - 2$$

Sample Data Set

(Graybill (1976), Theory and Application of the Linear Model, page 274)

x	y
.	2.10
1.0	4.81
1.1	3.60
1.3	4.90
1.6	3.05
1.8	3.44
1.8	3.17
1.8	3.34
2.1	1.61
2.4	1.22
2.6	0.20
2.6	1.56
2.7	0.55
2.9	-2.56
3.0	-0.34
3.5	-2.56
3.6	-2.96
4.1	-1.04
5.2	-4.64

Simple Linear Regression using sample data set.

Analysis of Variance					
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	119.36728	119.36728	107.25	<.0001
Error	16	17.80815	1.11301		
Corrected Total	17	137.17543			
Root MSE		1.05499	R-Square	0.8702	
Dependent Mean		0.96389	Adj R-Sq	0.8621	
Coeff Var		109.45167			
Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	6.99091	0.63288	11.05	<.0001
x	1	-2.40546	0.23228	-10.36	<.0001
Covariance of Estimates					
Variable	Intercept		x		
Intercept	0.4005369955		-0.135180861		
x	-0.135180861		0.0539524502		
CALIBRATION using SAS REGINV Macro					
Value of G Should be Less than 0.1					
Value of	Predicted				Value of
Dependent	Lower	Independent	Upper		
Variable	Bound	Variable	Bound		Alpha
2.1	1.07314	2.03325	2.99336		0.05
Standard Error of					
Independent	Critical Value				
Variable	from				
	t-distribution				
0.45290	2.11991				
	g				
	0.041903				

Using a contrast matrix compute the covariance matrix of the joint density of $(\hat{\alpha}, \hat{\beta})$.

$$\text{Cov}(\hat{\alpha}, \hat{\beta}) = \hat{\sigma}^2 \begin{pmatrix} 1 & -1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 & 0 \\ 0 & .3598686 & -.121455 \\ 0 & -.121455 & .0484744 \end{pmatrix} \begin{pmatrix} 1 & 0 \\ -1 & 0 \\ 0 & 1 \end{pmatrix}$$

$$= 1.11301 \begin{pmatrix} .3598686 & -.1214553 \\ -.1214553 & .0484744 \end{pmatrix}$$

$$\text{Cov}(\hat{\alpha}, \hat{\beta}) = \hat{\sigma}^2 \begin{pmatrix} V_{11} & V_{12} \\ V_{12} & V_{22} \end{pmatrix}$$

Output from Fieller macros written compute a 95% confidence interval of (1.03,2.99). Actual output follows:

FiellerM							
The Minimum Alpha (observed significance level) is 1.6845455E-8							
Critical Value							
a	b	df	alpha	from	Lower	Upper	
				t-distribution	Bound	ratio	Bound
-4.89091	-2.40546	16	0.05	2.11991	1.03150	2.03325	2.99369
Interval Details							
msg	anum	bnum	v11	v12	v22	v	
Assumptions are Met	-4.89091	-2.40546	1.35987	0.12146	0.048474	16	
alpha	s	stat0	stat	coef	df	t	m
0.05	1.05499	10.3561	1.6845E-8	0.975	16	2.11991	2.03325
1-k^2*				Under			
v22	A	AAI	B	C	Radical	LB	UB
0.95810	0.95810	0	-3.85652	2.95859	3.53428	1.03150	2.99369

FiellerF							
The Minimum Alpha (observed significance level) is 1.6845455E-8							
a	b	df	alpha	g	Lower	Upper	
					Bound	ratio	Bound
-4.89091	-2.40546	16	0.05	0.041903	1.03150	2.03325	2.99369
Interval Details							
a	b	m	s	v11	v12		
-4.89091	-2.40546	2.03325	1.05499	1.35987	0.12146		
v22	df	code_	code_g	LB	UB		
0.048474	16	disc	YES	1.03150	2.99369		

SAS Macros

REGINV written by SAS Institute allows for more significant digits since it uses output from proc glm in all of its computations. However, REGINV only applies to the specific case of calibration of the independent variable in a simple linear regression and hence is not as general in its application.

Two new macros were written to implement the use of Fieller's Theorem. They are more generic than the SAS macro reginv referenced above in that the code performs ratio estimation where the numerator and denominator follow a bivariate normal distribution. Inputs consist of the ratio, the covariance matrix, the mean square error and desired significance level. Diagnostics are given for cases where an acceptable interval does not exist. A slight difference in interval endpoints can be attributed to roundoff error.

The first macro is labeled FiellerM.sas (M=Milliken) and the second is called FiellerF.sas (F=Finney). While the formulas are slightly different both macros compute the same interval. The second macro is included for completeness since Finney's methodology is often referenced in the literature. A modified version of REGINV is included as reginv2.

The program Graybill.8.10.sas calls the two macros. All code was written using SAS versions 7 and 8 on a PC with Microsoft Windows 98 operating system. They are included as an attachment to this paper.

Acknowledgements

Draper and Smith (1998), Applied Regression Analysis -Third Edition, John Wiley & Sons, page 85 (3.28)

Finney, David J. (1978), Statistical Method in Biological Assay -Third Edition, Charles Griffin Company LTD

Graybill, Franklin A.(1976), Theory and Application of the Linear Model, Wadsworth & Brooks/Cole, Pacific Grove, California, Confidence Interval Formula is given on page 335

Miller, John Ph.D., George Mason University, Stat 656 Regression Analysis course notes

Milliken, George, Ph.D., Kansas State University, STAT 861 Linear Models II course notes, pages 8-34 to 8-37

SAS Principles of Regression Course Notes, Macro to produce inverse confidence intervals for individual values is found on pages 639-640, SAS Institute, Inc.: Cary, North Carolina

SAS Source Code

Graybill.8.10.sas Program which calls the macros.

```
/* SAS and FIELLER'S THEOREM, Graybill Problem 8.10 */
options nodate number center linesize=76 pagesize=54 pageno=1;
options replace probsig=1 firstobs=1 obs=max;
options mrecall label;
options xsync noxwait yearcutoff=1950;
title;
ods listing;

data Fieller;
input x y @@;
datalines;
.      2.10
1.0    4.81
1.1    3.60
1.3    4.90
1.6    3.05
1.8    3.44
1.8    3.17
1.8    3.34
2.1    1.61
2.4    1.22
2.6    0.20
2.6    1.56
2.7    0.55
2.9    -2.56
3.0    -0.34
3.5    -2.56
3.6    -2.96
4.1    -1.04
5.2    -4.64
;

data Fieller; set Fieller;
proc print; var x y; run;

proc glm data=Fieller;
model y = x / xpx inverse ss3;
ods output ParameterEstimates=ParameterEstimates;
ods output InvXpX=InvXpX;
ods output XpX=XpX;
run;

proc print data=XpX; title2 "XpX";run;
proc print data=InvXpX; title2 "InvXpX"; run;
proc print data=ParameterEstimates; title2 "Parameter Estimates"; run;

data XpX2; set XpX; n=_n_;
data XpX2; set XpX2; where n<3; drop parameter n;
title;

proc iml;
use ParameterEstimates;
read all var{Estimate StdErr tValue ProbT} into EstimateMatrix[ colname=vars
];
read all var{Estimate} into EstimateVector;
Estimate=t(EstimateVector);
print Estimate;
```

```

use XpX2;
read all var{intercept x} into XTX;
Cov=Inv(XTX);
contrast={1 -1 0,
          0 0 1};
Column={0,0}; Row={1 0 0};
CovarianceMatrix = Column || Cov;
CovarianceMatrix = Row // CovarianceMatrix;
prod=contrast*CovarianceMatrix*t(contrast);
print XTX, Cov, contrast, CovarianceMatrix, prod;
quit;

title 'FiellerM';
%FiellerM(Fieller,2.1-6.990912935,-2.405464142,1.3598686,.1214553,.0484744,16,
1.05499,.05);
title 'FiellerF';
%FiellerF(Fieller,2.1-6.990912935,-2.405464142,1.3598686,.1214553,.0484744,16,
1.05499,.05);

```

FiellerM Macro

```

%macro FiellerM(DataFile,a,b,v11,v12,v22,v,s,alpha);

/* (C) COPYRIGHT 2001 BY MARY A. MARION */

/* This macro produces a solution to Fieller's equations,
that is a (1-alpha)*100% confidence interval about an
estimate of the ratio Mu of two parameters from a simple
linear regression function y = alpha + Beta * x fitted to a
simple random sample of size n. Mu, is estimated by a/b.
The errors are assumed to be normally distributed.

Written by Mary A. Marion while a student at
Kansas State University April 12,2000.
See class notes pages 8-34.

Input Parameters:

DataFile containing x and y
(independent and dependent variables)

Regression Parameter Estimates
a =  $y_0 - \hat{\alpha}$  where  $\hat{\alpha}$  = Intercept in regression equation
b =  $\beta$  Slope from the regression equation

Covariance Matrix of regression parameter estimates
V11 Var(a)
V12 Cov(a,b)
V22 Var(b)
v Degrees of freedom = n-2 ( n = number of observations )
s MSE from Simple Linear Regression
alpha Significance level

```

Acknowledgements:

Draper and Smith (1998), Applied Regression Analysis -Third Edition,

John Wiley & Sons, page 85 (3.28).

Finney, David J. (1978), Statistical Method in Biological Assay -Third Edition, Charles Griffin Company LTD.

Graybill, Franklin A.(1976),
Theory and Application of the Linear Model
Wadsworth & Brooks/Cole
Pacific Grove, California
Confidence Interval Formula is given on page 335

Miller, John Ph.D., George Mason University
Stat 656 Regression Analysis course notes.

Milliken, George, Ph.D., Kansas State University
STAT 861 Linear Models II course notes, pages 8-34 to 8-37.

SAS Principles of Regression Course Notes, Page 639-640,
SAS Institute, Inc.: Cary, North Carolina.

Algorithms are from course notes referenced above.

```
'Let k = s * t(alpha/2,v) / b.'  
'When 1 - k^2 * v22 > 0, model assumptions are true.'  
'A Confidence Interval Does not Exist if 1- k^2 * v22 le 0.'  
'The Minimum Alpha (observed significance level) is ' stat;
```

Assumed is a quadratic equation of the form
 $a x^2 + b x + c = 0$. a is constrained to be
positive in order for this to work. Hence code
includes a test for negative a. */

```
/* options nomprint nosource; */
```

```
options formdlm='';  
%local AA UnderRad a b v11 v12 v22 v s alpha;  
data interval; length msg $ 19;  
msg='';  
anum=&a; bnum=&b;  
v11=&v11; v12=&v12; v22=&v22;  
v=&v; alpha=&alpha; s=&s;  
alpha=&alpha;  
  
data interval; set interval; file print;  
stat0 = &b / (&s*sqrt(v22)); stat0=abs(stat0);  
stat= 1 - probt(stat0,&v) ; stat=stat*2;  
put / ' The Minimum Alpha (observed significance level) is ' stat;  
run;  
  
data interval; set interval;  
coef=1-&alpha/2;  
df=&v;  
t=tinv(coef,df);  
m=anum/bnum ;  
k= (s*t) / bnum;  
denominator=1-k*k*v22;  
AA = 1.0 - (k*k*v22);  
AAI=floor(AA);  
call symput('AAI',trim(left(put(AAI,8))));
```

```

B = (2*k*k*v12)-2*m;
C = m*m- (k*k*v11);
UnderRad = B**2 - 4*AA * C;
call symput('UnderRad',trim(left(put(UnderRad,16.8))));
run;

%if %eval(&AAI) < 0 %then %do;
/* The if statement checks to see whether 1-k*k*v22
   is le 0. This is same as seeing if AAI < 0 */
data interval; set interval;
  msg = 'Assumptions not Met' ;
  options formdlim=' '; title1 ' ';
  proc print data=interval noobs label;
  title2 "Interval Diagnostics";
  label UnderRad='Under Radical' denominator='1-k^2*v22' AA='A';
  run; title2;
%end; %else %do;
data interval; set interval;
  if AA lt 0 then do;
    AA = -AA; B = -B; C=-C; end;
    LB = ( -B - sqrt(B**2-4*AA*C) ) / (2*AA);
    UB = ( -B + sqrt(B**2-4*AA*C) ) / (2*AA);
    msg='Assumptions are Met'; run;
    options formdlim=' ';
  proc print data=interval noobs label;
  var anum bnum df alpha t LB m UB ;
  label anum='a' bnum='b'
  m=ratio LB='Lower Bound' UB='Upper Bound'
  t='Critical Value from t-distribution'; run;
  proc print data=interval noobs label; title2 "Interval Details";
  label UnderRad='Under Radical' denominator='1-k^2*v22' AA='A';
  run; title2;
%end;

title2;
proc datasets; delete interval; run &cancel;
options mprint source;
%mend FiellerM;

```

FiellerF macro

```

%macro FiellerF(DataFile,a,b,v11,v12,v22,v,s,alpha);

/* (C) COPYRIGHT 2001 BY MARY A. MARION */

/* This macro produces a solution to Fieller's equations,
   that is a (1-alpha)*100% confidence interval about an
   estimate of the ratio Mu of two parameters from a simple
   linear regression function y = alpha + Beta * x fitted to a
   simple random sample of size n. Mu, is estimated by a/b.
   The errors are assumed to be normally distributed.

   Written by Mary A. Marion while a student at
   Kansas State University April 12,2000.
   See class notes pages 8-34.

   Input Parameters:

```

```

DataFile containing x and y (independent and dependent variables)

Regression Parameter Estimates
a =  $y_0 - \hat{\alpha}$  where  $\hat{\alpha}$  = Intercept in regression equation
b =  $\beta$  Slope from the regression equation

Covariance Matrix of regression parameter estimates
V11 Var(a)
V12 Cov(a,b)
V22 Var(b)
v Degrees of freedom = n-2 ( n = number of observations )
s MSE from Simple Linear Regression
alpha Significance level

Acknowledgements:

Draper and Smith (1998), Applied Regression Analysis -Third Edition,
John Wiley & Sons, page 85 (3.28).

Finney, David J. (1978), Statistical Method in Biological Assay -Third
Edition, Charles Griffin Company LTD.

Graybill, Franklin A.(1976),
Theory and Application of the Linear Model
Wadsworth & Brooks/Cole
Pacific Grove, California
Confidence Interval Formula is given on page 335

Miller, John Ph.D., George Mason University
Stat 656 Regression Analysis course notes.

Milliken, George, Ph.D., Kansas State University
STAT 861 Linear Models II course notes, pages 8-34 to 8-37.

SAS Principles of Regression Course Notes, Page 639-640,
SAS Institute, Inc.: Cary, North Carolina.

Algorithms are from course notes referenced above.
'Let k = s * t(alpha/2,v) / b.'
'When 1 - k^2 * v22 > 0, model assumptions are true.'
'A Confidence Interval Does not Exist if 1- k^2 * v22 le 0.'
'The Minimum Alpha (observed significance level) is ' stat;

/* options nomprint nosource; */
options formdlm='';
%local a b v11 v12 v22 v s alpha;

data interval; length msg $ 19;
msg='';
anum=&a; bnum=&b;
v11=&v11; v12=&v12; v22=&v22;
v=&v; alpha=&alpha; s=&s;
alpha=&alpha;
coef=1-&alpha/(2);
df=&v;

data interval; set interval; file print;
stat0 = &b / (&s*sqrt(v22)); stat0=abs(stat0);

```

```

stat= 1 - probt(stat0,&v) ; stat=stat*2;
put / '          The Minimum Alpha (observed significance level) is ' stat;
run;

data interval; set interval;
/* Equation 4.12.3 */
t=tinv(coef,df);
g = t*t * s*s * v22 / (bnum*bnum);
call symput('g',trim(left(put(g,15.8)))));
m=anum/bnum ;

/* Equation 4.12.3 */
disc=(v11-2*m*v12+m*m*v22-g*(v11-(v12**2)/v22));
if disc<0 then code_disc='NO '; else code_disc="YES";
if disc<0 then k=.; else k=sqrt(disc);
dif= ( (t*s) / ( abs(bnum) ) ) * k ;
val=m-g*v12/v22;
LB=(val-dif)/(1-g);
UB=(val+dif)/(1-g);

if LB>UB then do;
dummy=LB; LB=UB; UB=dummy; end;

code_g='YES'; msg='Assumptions are Met';
if g ge 1 then do;
code_g='NO '; msg = 'Assumptions not Met' ; end;
if g ge 1 and disc<0 then LB=.;
if g ge 1 and disc<0 then UB=.;

options formdlm=' '; title;
proc print data=interval noobs label;
var anum bnum df alpha g LB m UB ;
label anum='a' bnum='b'
m=ratio LB='Lower Bound' UB='Upper Bound'
t='Critical Value from t-distribution'; run;

proc print data=interval noobs label;
title2 "Interval Details";
label anum='a' bnum='b';
var anum bnum m s v11 v12 v22 df code_disc code_g LB UB;
run; title2;

proc datasets; delete interval; run cancel;
options mprint source;
%mend FiellerF;

```

REGINV2 macro for restricted calibration problems.

```

%macro reginv2(olddata,x,y,alpha,resids);
/* This macro produces inverse confidence intervals for individual
values for the simple linear regression problem found in dataset
olddata.
Text: SAS Principles of Regression Course Notes, Page 639-640
Text: Draper and Smith (1998),
Applied Regression Analysis -Third Edition,
page 85 (3.28)
Text: Graybill, Franklin A. (1976),

```



```

Theory and Application of the Linear Model */
options nomprint nosource;
proc reg data=&olddata outest=estimate
  (keep=_rmse_ &x intercept
  rename=(&x=b1 intercept=b0));
  model &y=&x /covb;
  output out=&resids
    p=p r=r press=press
    l95=l95 u95=u95
    l95m=l95m u95m=u95m
    stdp=stdp stdi=stdi stdr=stdr h=h
    cookd=cookd covratio=covratio
    dffits=dffits
    student=student rstudent=rstudent;
  run;
proc contents data=estimate; title2 'estimate'; run cancel;
proc print data=estimate; title2 'estimate'; run cancel;
proc means noprint data=&olddata;
  var &x;
  output out=means(keep=xmean n css)
    mean=xmean n=n css=css;
  run;
/* proc print data=means; title 'means'; run; */
data combine(drop=_rmse_);
  set estimate; set means;
  mse=_rmse_*_rmse_;
/* proc print data=combine;
  title1 'Approximate Confidence Intervals';
  title2 'Summary Information for Calculations';
  title3 'combine';
  run; */
data final(drop=&x);
  if _n_=1 then set combine;
  set &olddata;
  if &x=.;
  /* Graybill, page 277 (8.5.5) */
  xpred=(&y-b0)/b1;
/* proc print data=final; run; */
data interval; set final;
  a1=1/n;
  diff=(xpred-xmean)**2;
  a2=diff/css;
  mult=mse/b1**2;
  var=mult*(1+a1+a2);
  stderr=sqrt(var);
  df=n-2;
  coef=1-&alpha/2;
  t=tinv(coef,df);
  upper=xpred+t*stderr;
  lower=xpred-t*stderr;
  g=(mse*t**2)/(css*b1**2);
  alpha=&alpha;
proc print data=interval noobs label;
  var &y lower xpred upper alpha stderr t g;
  label &y = 'Value of Dependent Variable'
    upper='Upper Bound'
    lower='Lower Bound'
    xpred='Predicted Value of Independent Variable'

```

```

        alpha='Value of Alpha'
        stderr='Standard Error of Independent Variable'
        t='Critical Value from t-distribution';
    title3 'CALIBRATION';
    title4 'Value of G Should be Less than 0.1';
run;
title3; title4;
proc datasets; delete combine estimate final; run;
options mprint source;
%mend reginv2;

```

Program to run reginv2 for use with restricted calibration problems.

```

/* Graybill, Linear Models Problem 8.10, page 336, data is on page 274 */
options nodate number center linesize=76 pagesize=54 pageno=1;
options replace probsig=1 firstobs=1 obs=max;
options mrecall label;
options xsync noxwait;
options mlogic mprint symbolgen formdlm='' label;

data olddata;
infile "c:\Fieller.dat";
input x y;
data addin;
input x y;
datalines;
. 2.10
;
data olddata; set olddata addin; run;
proc sort; by x; run;
proc print;run;

%reginv2(olddata,x,y,.05,resids);

```

SAS OUTPUT

Obs	x	y
1	.	2.10
2	1.0	4.81
3	1.1	3.60
4	1.3	4.90
5	1.6	3.05
6	1.8	3.44
7	1.8	3.17
8	1.8	3.34
9	2.1	1.61
10	2.4	1.22
11	2.6	0.20
12	2.6	1.56
13	2.7	0.55
14	2.9	-2.56
15	3.0	-0.34
16	3.5	-2.56
17	3.6	-2.96
18	4.1	-1.04
19	5.2	-4.64

The GLM Procedure

Number of observations 19

NOTE: Due to missing values, only 18 observations can be used in this analysis.

The GLM Procedure

The X'X Matrix

	Intercept	x	y
Intercept	18	45.1	17.35
x	45.1	133.63	-6.152
y	17.35	-6.152	153.8989

The GLM Procedure

X'X Inverse Matrix

	Intercept	x	y
Intercept	0.3598685805	-0.121455309	6.9909129346
x	-0.121455309	0.0484744028	-2.405464142
y	6.9909129346	-2.405464142	17.808145181

The GLM Procedure

Dependent Variable: y

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	119.3672826	119.3672826	107.25	<.0001
Error	16	17.8081452	1.1130091		
Corrected Total	17	137.1754278			

R-Square	Coeff Var	Root MSE	y Mean
0.870180	109.4517	1.054992	0.963889

Source	DF	Type III SS	Mean Square	F Value	Pr > F
x	1	119.3672826	119.3672826	107.25	<.0001

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	6.990912935	0.63287992	11.05	<.0001
x	-2.405464142	0.23227667	-10.36	<.0001

XpX

Obs	Parameter	Intercept	x	y
1	Intercept	18	45.1	17.35
2	x	45.1	133.63	-6.152
3	y	17.35	-6.152	153.8989

InvXpX

Obs	Parameter	Intercept	x	y
1	Intercept	0.3598685805	-0.121455309	6.9909129346
2	x	-0.121455309	0.0484744028	-2.405464142
3	y	6.9909129346	-2.405464142	17.808145181

Parameter Estimates

Obs	Dependent	Parameter	Estimate	StdErr	tValue	Probt
1	y	Intercept	6.990912935	0.63287992	11.05	<.0001
2	y	x	-2.405464142	0.23227667	-10.36	<.0001

ESTIMATE

6.9909129 -2.405464

XTX

18 45.1
45.1 133.63

COV

0.3598686 -0.121455
-0.121455 0.0484744

CONTRAST

1 -1 0
0 0 1

COVARIANCEMATRIX

1 0 0
0 0.3598686 -0.121455
0 -0.121455 0.0484744

PROD

1.3598686 0.1214553
0.1214553 0.0484744

FiellerM

10

The Minimum Alpha (observed significance level) is 1.6845455E-8

FiellerM

11

a	b	df	alpha	Critical Value from t-distribution	Lower Bound	ratio	Upper Bound
-4.89091	-2.40546	16	0.05	2.11991	1.03150	2.03325	2.99369

FiellerM

12

Interval Details

msg	anum	bnun	v11	v12	v22	v
Assumptions are Met	-4.89091	-2.40546	1.35987	0.12146	0.048474	16

alpha	s	stat0	stat	coef	df	t	m	k
0.05	1.05499	10.3561	1.6845E-8	0.975	16	2.11991	2.03325	-0.92975
1-k^2*						Under		
v22	A	AAI	B	C		Radical	LB	UB
0.95810	0.95810	0	-3.85652	2.95859		3.53428	1.03150	2.99369

FiellerF 13

The Minimum Alpha (observed significance level) is 1.6845455E-8

14

a	b	df	alpha	g	Lower Bound	ratio	Upper Bound
-4.89091	-2.40546	16	0.05	0.041903	1.03150	2.03325	2.99369

15

Interval Details

a	b	m	s	v11	v12
-4.89091	-2.40546	2.03325	1.05499	1.35987	0.12146
v22	df	code_ disc	code_g	LB	UB
0.048474	16	YES	YES	1.03150	2.99369

REGINV.8.10.LST

The SAS System

1

Obs	x	y
1	.	2.10
2	1.0	4.81
3	1.1	3.60
4	1.3	4.90
5	1.6	3.05
6	1.8	3.44
7	1.8	3.17
8	1.8	3.34
9	2.1	1.61
10	2.4	1.22
11	2.6	0.20
12	2.6	1.56
13	2.7	0.55

14	2.9	-2.56
15	3.0	-0.34
16	3.5	-2.56
17	3.6	-2.96
18	4.1	-1.04
19	5.2	-4.64

The SAS System

2

The REG Procedure
Model: MODEL1
Dependent Variable: y

Analysis of Variance

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	119.36728	119.36728	107.25	<.0001
Error	16	17.80815	1.11301		
Corrected Total	17	137.17543			

Root MSE	1.05499	R-Square	0.8702
Dependent Mean	0.96389	Adj R-Sq	0.8621
Coeff Var	109.45167		

Parameter Estimates

Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	6.99091	0.63288	11.05	<.0001
x	1	-2.40546	0.23228	-10.36	<.0001

Covariance of Estimates

Variable	Intercept	x
Intercept	0.4005369955	-0.135180861
x	-0.135180861	0.0539524502

The SAS System
estimate

3

CALIBRATION

Value of G Should be Less than 0.1

Value of Dependent Variable	Lower Bound	Predicted Value of Independent Variable	Upper Bound	Value of Alpha
-----------------------------------	----------------	--	----------------	-------------------

2.1	1.07314	2.03325	2.99336	0.05
-----	---------	---------	---------	------

Standard Error of Independent Variable	Critical Value from t-distribution	g
---	--	---

0.45290	2.11991	0.041903
---------	---------	----------